# Maintainer: Antonio Rojas <arojas@archlinux.org>

pkgname=jupyter-retrolab
_pipname=retrolab
pkgver=0.3.21
pkgrel=2
pkgdesc='JupyterLab Distribution with a retro look and feel'
arch=(any)
url='https://github.com/jupyterlab/retrolab'
license=(BSD)
makedepends=(python-build python-installer)
depends=(jupyterlab)
source=(https://pypi.io/packages/source/${_pipname:0:1}/$_pipname/$_pipname-$pkgver.tar.gz)
sha256sums=('06df283b25ba52678d679b2a64d925ece3e14e38dff1094d973fae2af15ca0a6')

build() {
  cd $_pipname-$pkgver
  python -m build --wheel --no-isolation
}

package() {
  cd $_pipname-$pkgver
  python -m installer --destdir="$pkgdir" dist/*.whl

  mv "$pkgdir"/{usr/,}etc
  install -Dm644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
}
